Université de Nice
Département de Mathématiques

L3MASS et L3 Maths, année 2012-2013

NOM : Date : .
PRENOM : Salle et heure : .

**Feuille-question du TP 7**
**Dynamique des populations : matrices de Leslie**

Considérons une population de saumons en limitant nos observations aux seules femelles. Supposons qu'elles vivent au maximum 3 ans, avec un taux de survie de 53% la première année et 22% la seconde, et enfin supposons que chaque femelle donne naissance à 4 juveniles au cours de sa deuxième année et à 5 au cours de sa troisième année. Si l'on désigne respectivement par $j_t$, $p_t$ et $a_t$ les effectifs à l'instant $t$ des femelles juvéniles, des femelles préadultes (saumons de 1 an) et des femelles adultes (saumons de 2 ans), les informations précédentes peuvent s'écrire :

$$\left\{\begin{array}{lcl} j_{t+1} & = & 4p_t + 5a_t \\ p_{t+1} & = & 0,53j_t \\ a_{t+1} & = & 0,22p_t \end{array}\right. \qquad (1)$$

**1.-** Indiquer qu'elle est la matrice de Leslie $L$ de cette dynamique, calculer $L^2$ et $L^3$ puis vérifier que la matrice `L` est *primitive.*

**2.-** Le théorème de Perron Frobenius assure que la distribution limite est un vecteur propre associé à la valeur propre de plus grand module, dite *valeur propre dominante.* Nous allons expérimenter ceci pour les saumons considérés. Comme c'est une distribution, il convient de diviser les coefficients du vecteur propre donné par `Scilab` (qui est normé pour la norme euclidienne) par la somme de ses coefficients (ils sont positifs) obtenant ainsi un vecteur normé pour la norme des la somme des (valeurs absolues) des coefficients, souvent appelée “norme 1”.

```
//Calcul des vecteurs propres et valeurs propres
[vcp,vlp]=spec(L) ;
//vecteur ligne ('r'=row) des sommes par colonnes
normes1vcp=sum(vcp,'r') ;
//calcul des vecteurs "renormalisés"
for ligne=1:3 ;
      vcpnorme(ligne,:)=vcp(ligne,:)./normes1vcp ;
end ;
disp(vcpnorme,"vecteurs propres=", vlp,"valeurs propres") ;
```

Quelle valeur propre dominante $\lambda$ trouvez-vous et quelle distribution limite $V_\infty$ ?

**3.-** On veut calculer la dynamique d'une population qui, à l'instant initial, comporte 2000 juvéniles, 1000 préadultes, et 500 adultes.

```
//condition initiale
V0=[2000 ;1000 ;500] ;
// il y aura tMax+1 instants
tMax=20 ;
// initialisation de V
V=zeros(3,tMax+1) ;
V(:,1)=V0 ;
```

```
//voici les instants où V(t) sera calculé
tt=0:tMax;
//Calcul des V_{t+1} pour t>0 ; rangé dans V(:,t+1+1)
for t=0:tMax-1;//on calcule V_{t+1}
V(:,t+1+1)=L*V(:,t+1);
end;
//size(tt)//pour debuggage: verifier que les tailles sont egales.
//size(V)
j=V(1,:);//les juveniles
p=V(2,:);//les preadultes
a=V(3,:);//les adultes
//juveniles en vert, preadultes en bleu, adultes en rouge
plot(tt,j,'g-*');plot(tt,p,'b--o');plot(tt,a,'r.->');
```

Esquissez avec soin les tracés obtenus

**4.-** Dynamique de la distribution des trois sous populations

```
xset("window",1);
total=j+p+a;
plot(tt,j./total,'g-*'); plot(tt,p./total,'b--o'); plot(tt,a./total,'r.->');
```

Qu'observez-vous ? Pouviez-vous le prévoir ?

**5.-** Log de cette dynamique

```
xset("window",2);
plot(tt,log(j),'g-*'); plot(tt,log(p),'b--o'); plot(tt,log(a),'r.->');
```

Qu'observez-vous ? Expliquez en quoi cela montre que la dynamique des trois population est "en $Ce^{\lambda t}$" ; quelle est la valeur de $\lambda$ ?

**6.- Une autre manière de programmer le tracé la dynamique**

```
xset("window",3);
```

```
Couleur=['g-*','b--o','r.->'] ; for ligne=1:3 ;
W(ligne,:)=V(ligne,:)./total ;
plot(tt,W(ligne,:),Couleur(ligne))
end ;
```

**7.-** Un modèle de Leslie a été proposé pour représenter la dynamique de la population d'un pays. Ne prenant en compte que les individu de sexe féminin, c'est-à-dire en ignorant les naissances masculines dans les taux de fécondités des classes, on a choisi dix classes d'age d'une durée de cinq ans et un pas de temps de cinq ans également. On a obtenu les coefficients suivants sur la première ligne de la matrice

$$\begin{pmatrix} 0,000 & 0,0010 & 0,878 & 0,3487 & 0,4761 & 0,3377 & 0,1833 & 0,0761 & 0,174 & 0,0010 \end{pmatrix}$$

et les coefficients suivants sur la sous diagonale

$$\begin{pmatrix} 0,9966 & 0,9983 & 0,9979 & 0,9968 & 0,9961 & 0,9947 & 0,9923 & 0,9987 & 0,9831 \end{pmatrix}$$

Saisir la matrice de Leslie $L$ correspondante en commençant par

`LL=zeros(10,10) ;`
`LL(1,2)=0.001 ;`...........
pour la première ligne, puis par
`LL(2,1)=0.9966 ;` .......
pour la sous diagonale et executer le code suivant :

```
[vcpLL,vlpLL]=spec(LL) ;
//NB: ouvrir en grand la fenetre "console" avant d'executer les instructions suivantes
disp(LL,'Matrice de Leslie LL =') ;
disp(vlpLL,'valeurs propres de LL =') ;
disp(vlpLL*conj(vlpLL),'carré des normes euclidiennes') ;
disp(vcpLL,'vecteurs propres de LL =') ;
vlpLL(1,1)^(1/5) ;
```

Quelle est la valeur propre la plus grande en norme euclidienne ?

**8.-** Quel est le taux de croissance annuel de cette population ?

**9.-** Quelle est la distribution d'équilibre de cette population ? Donner le code `Scilab`.

Révision : répondre aux questions suivantes *sans utiliser* `Scilab`.

**10a.-** Comment s'écrit, en `Scilab`, le vecteur colonne de composantes 1, 2, 3,

**10b-** Comment s'écrit le vecteur ligne ayant ces mêmes composantes ?

**10c.-** Que représente `A(2,:)` ?

**10d.-** Que représente B( :,3) ?

**10e.-** Comment calculer la matrice des vecteurs propres d'une matrice `A` ?

**10f.-** Ces vecteurs propres sont-ils en ligne ou en colonne ?

**10g.-** Si `plot(0:0.1:2,y,'r.x')` ne retourne pas d'erreur que retourne la commande `size(y)` ?

**10h.-** Que retourne la commande `disp('décembre',25,'Noel a lieu le ');` ?

**10i.-** Si `A=[1,2,3;4,5,6;7,8,9]` que retourne `sum(A,'c')` ?

**10j.-** Est-ce un vecteur ligne ou un vecteur colonne ?

**10j.-** Comment tracer le graphe de la fonction $f(x) = x^2 - 1$ pour $x$ entre 0 et 1 ?

**10j.-** Comment définir un champ de vecteurs $w$ pour que `ode(M0,0,tt,w)` trace la solution issue du point $M_0$ du système différentiel $x' = y$, $y' = x$ pour $tt$ entre 0 et 5 ?

**10j.-** Comment tracer la solution calculée de cette façon ?